



Technical Note 04-15

SOAP でピクチャを返す

By David Adams
Technical Note 04-15

(原題: Returning Pictures Through SOAP)

概要

4D バージョン 2003 以降は、メソッドを Web サービスとして公開することが可能になりました。Web サービスのリクエストおよびレスポンスは、内部的には XML 形式の SOAP (Simple Object Access Protocol) メッセージです。4thDimension は、サポートするデータタイプを自動的に変換して SOAP で返すことができますが、ピクチャだけは SOAP でそのまま返すことができません。とはいえ BLOB を SOAP に含めることは可能なので、PICTURE TO BLOB コマンドを活用すれば、この問題を解決することができます。

Web サービス+ピクチャには次のような可能性を秘めています。

ダイナミックあるいはスタティックな画像を含む Web ページを SOAP で配信
4D データベースをイメージアーカイブまたはライブラリ、配信ハブとして利用

この Tech Note では、SOAP レスポンスにピクチャを組み込む方法を取り上げます。サンプルデータベースには SOAP ベースイメージサーバの簡単なコードが含まれています。本文中、実際に応用する際に役立つヒントが提案として挙げられています。

注記:

有用な情報は、4th Dimension ランゲージリファレンスの Web サービスの項、および Tech Note 04-12 にも掲載されています。

サンプルで取り上げている機能

SOAP リクエストに応答する
利用可能な画像の一覧を提供する
利用可能な画像形式の一覧を提供する
画像をロード、変換、配信する

SOAP リクエストに応答する

メソッドを Web サービスで公開するためには、次の条件を満たさなくてはなりません。

4D の Web サーバが起動している

「環境設定/Web サービス/Web サービスリクエストを許可する」が有効である

「対象メソッドのプロパティ/Web サービスとして提供する」が有効である

利用可能な画像の一覧を提供する

サンプルデータベースでは、以下のようなテーブルに画像を登録しています。今回はサンプルなので Name をそのままユニークキーとして利用しています。現実のシステムでは、重複不可で変更不可の倍長整数か文字列が望ましいでしょう。

GetPictureList

メソッド GetPictureList は、利用可能な画像の Name を返す Web サービスです。

```
ALL RECORDS([Images])
ORDER BY([Images];[Images]Name)
ARRAY STRING(80;getPicture_ImageNames_as;0)
SELECTION TO ARRAY([Images]Name;getPicture_ImageNames_as)
UNLOAD RECORD([Images])
SOAP DECLARATION(getPicture_ImageNames_as;String array ;SOAP Output ;"outPictureNames")
```

現実のシステムでは、サイズ、作者などのクエリ条件を引数として渡すようなコードが便利かもしれません。

利用可能な画像形式の一覧を提供する

4th Dimension は、内部的に PICT 形式で画像を保存しています。MacOS のフォーマットである PICT 形式には、多くのブラウザが非対応となっています。4D は PICT を GIF に変換することができ、また Quicktime4 以降がインストールされていれば、Quicktime の対応形式すべてに変換することができます。PICTURE TYPE LIST コマンドは利用可能な画像形式の一覧を返しますので、これを Web サービスで返します。中央のサーバに Quicktime がインストールされていれば、SOAP クライアント側は希望の形式で画像を受け取ることができるので Quicktime は不要です。

GetPictureFormatList

メソッド GetPictureFormatList は、スタートアップで作成された MIME タイプのインタープロセス配列をプロセス配列にコピーしています。SOAP レスポンスでは、\$0、プロセス変数、プロセス配列の値しか返せないのが、インタープロセス配列はこのようなコピーして使用します。

```
ARRAY TEXT(getPictureFormatMimeTypes_at;0)
COPY ARRAY(<>pictureFormatMimeTypes_at;getPictureFormatMimeTypes_at)
SOAP DECLARATION(getPictureFormatMimeTypes_at;Text array;
SOAP Output;"outSupportedMIMETypes")
```

スタートアップでこの配列を作成するのはメソッド `pictureInitializeFormatCodes` です。
`PICTURE TYPE LIST` コマンドは、MIME タイプではなく Quicktime のタイプコードを返すので、テーブル `[Picture_Format_Codes]` の対応表に従ってこれを MIME に変換しています。

画像をロード、変換、配信する

`getPicture` メソッドには、引数として画像の名前と MIME タイプを渡します。クライアントは事前に `GetPictureList` および `GetPictureFormatList` メソッドを Web サービスコールすることにより、必要な情報を取得することができます。

クライアント側の処理: 画像を展開する

SOAP メッセージに含まれた BLOB は Base64 エンコードを施されています。これを元に画像に戻す処理は 4D のコマンドを使えば非常に簡単です。

```
C_TEXT($1;$name_t)
C_TEXT($2;$mimeType_t)
C_POINTER($3;$picture_p)

$name_t:=$1
$mimeType_t:=$2
$picture_p:=$3

SET WEB SERVICE PARAMETER("inPictureName";$name_t)
SET WEB SERVICE PARAMETER("inPictureMimeType";$mimeType_t)
CALL WEB SERVICE(requestGetAccessURL ;
"ReturningPicturesWithSoap#getPicture"; "getPicture";"http://www.4d.com/namespace/default";Web
Service Dynamic )

If (OK=1)
    C_BLOB($pictureBlob_x)
    GET WEB SERVICE RESULT($pictureBlob_x;"outPictureBlob";*)
    BLOB TO PICTURE($pictureBlob_x;$picture_p->)
End if
```

以上のプロキシメソッドをコールすれば、画像をピクチャ変数に受け取ることができます。

```
C_PICTURE(rhino_picture)
request_getPicture ("Baby White Rhino";"image/pict";->rhino_picture)
```

補足情報を取得する

サンプルのデータベースでは、画像以外に様々なデータ（サイズ、作者、説明 etc）がフィールドに登録されていました。Web サービスでは、それぞれの情報を取得するために個別のメソッドを用意したほうがクライアントに柔軟性を持たせることができ便利です。

<code>getPictureData</code>	画像の説明、サイズ、幅、高さ、作者を返す
<code>getPictureListExtended</code>	画像と共に説明、サイズ、幅、高さ、作者を返す

getPictureWithType
getThumbnail

画像と共に MIME タイプを返す
画像のサムネイルを生成して返す

サンプルデータベース




Image Browser

Web サービス（この場合は自身）サーバから画像の一覧とサムネイルを取得するクライアント




Image Conversion

Web サービス（この場合は自身）サーバから形式を変換された画像を取得するクライアント

追記

サンプルでは、ピクチャと他のデータを同じレコードに登録しました。実際のシステムでは、大きな画像は外部ファイルとして保存し、データベースにはパスだけを登録したほうが効率的です。入力データに基づいてダイナミックに画像を生成するには 4D Chart や 4D Draw のようなプラグインが便利です。

ソース	使用コマンド例
4D Chart	CT Area to picture
ディスク上のファイル	Open document+RECEIVE PACKET DOCUMENT TO BLOB READ PICTURE FILE LOAD COMPRESS PICTURE FROM FILE
ネットワーク上のファイル	TCP_Open+TCP_ReceiveBLOB
PICT リソース	GET PICTURE RESOURCE
ピクチャライブラリ	GET PICTURE FROM LIBRARY
レコード	-